<commit_before>{ stdenv, fetchurl, alsaLib, faac, libjack2, lame, libopus, libpulseaudio, libsamplerate, libvorbis }:

with stdenv.lib;

stdenv.mkDerivation rec {
  name = "darkice-${version}";
  version = "1.2";

  src = fetchurl {
    url = "mirror://sourceforge/darkice/${version}/darkice-${version}.tar.gz";
    sha256 = "0m5jzmja7a9x15zl1634bhxrg3rccph9rkar0rmz6wlw5nzakyxk";
  };

  configureFlags = [
    "--with-alsa-prefix=${alsaLib}"
    "--with-faac-prefix=${faac}"
    "--with-jack-prefix=${libjack2}"
    "--with-lame-prefix=${lame}"
    "--with-opus-prefix=${libopus}"
    "--with-pulseaudio-prefix=${libpulseaudio}"
    "--with-samplerate-prefix=${libsamplerate}"	
    "--with-vorbis-prefix=${libvorbis}"
#    "--with-aacplus-prefix=${aacplus}" ### missing: aacplus
#    "--with-twolame-prefix=${twolame}" ### missing: twolame
  ];

  meta = {
    homepage = http://darkice.org/;
    description = "Live audio streamer";
    license = stdenv.lib.licenses.gpl3;
    maintainers = with stdenv.lib.maintainers; [ ikervagyok ];
  };
}<commit_msg>darkice: Fix build with opus and vorbis support<commit_after>{ stdenv, buildEnv, fetchurl, alsaLib, faac, libjack2, lame, libogg, libopus, libpulseaudio, libsamplerate, libvorbis }:

let
  oggEnv = buildEnv {
    name = "env-darkice-ogg";
    paths = [
      libopus libvorbis libogg
    ];
  };

in stdenv.mkDerivation rec {
  name = "darkice-${version}";
  version = "1.2";

  src = fetchurl {
    url = "mirror://sourceforge/darkice/${version}/darkice-${version}.tar.gz";
    sha256 = "<KEY>";
  };

  configureFlags = [
    "--with-alsa-prefix=${alsaLib}"
    "--with-faac-prefix=${faac}"
    "--with-jack-prefix=${libjack2}"
    "--with-lame-prefix=${lame}"
    "--with-opus-prefix=${oggEnv}"
    "--with-pulseaudio-prefix=${libpulseaudio}"
    "--with-samplerate-prefix=${libsamplerate}"	
    "--with-vorbis-prefix=${oggEnv}"
#    "--with-aacplus-prefix=${aacplus}" ### missing: aacplus
#    "--with-twolame-prefix=${twolame}" ### missing: twolame
  ];

  meta = {
    homepage = http://darkice.org/;
    description = "Live audio streamer";
    license = stdenv.lib.licenses.gpl3;
    maintainers = with stdenv.lib.maintainers; [ ikervagyok ];
  };
}
{'fraction_non_alphanumeric': 0.1892502258355917, 'fraction_numerical': 0.014001806684733513, 'mean_word_length': 3.483805668016194, 'pattern_counts': {'":': 0, '<': 4, '<?xml version=': 0, '>': 4, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1769551', 'n_tokens_mistral': 848, 'n_tokens_neox': 822, 'n_words': 123}